Shifted Partial Quadratics, their Pushforwards, and Signature Invariants for Tangles

Abstract. Following a general discussion of the computation of zombians of unfinished columbaria (with examples), I will tell you about my recent joint work w/ Jessica Liu on what we feel is the "textbook" extension of knot signatures to tangles, which for unknown reasons, is not in any of the textbooks that we know.

Jessica Liu

Columbaria in an East Sydney Cemetery

Image: Freepik.com

Jacobian, Hamiltonian, Zombian

Prior Art on signatures for tangles / braids. Gambaudo and Ghys [GG], Cimasoni and Conway [CC], Conway [Co], Merz [Me]. All define signatures of tangles / braids by first closing them to links and then work hard to derive composition properties.

Why Tangles? • Faster!

• Conceptually clearer proofs of invariance (and of skein relations).

• Often fun and consequential:



$2^{n/2}+2^{n/2}+2^{\sqrt{n}} \ll 2^n$

∘ The Jones Polynomial ⇝ The Temperley-Lieb Algebra.

∘ Khovanov Homology ⇝ "Unfinished complexes", complexes in a category.

∘ The Kontsevich Integral ⇝ Associators.

∘ HFK ⇝ OMG, type D, type A, $\mathcal{A}_\infty$, ...

Knot-tease



Zombies: Freepik.com

Computing Zombians of Unfinished Columbaria.

• Must be no slower than for finished ones.

• Future zombies must be able to complete the computation.

• Future zombies must not even know the size of the task that today's zombies were facing.

• We must be able to extend to ZPUCs, Zombie Processed Unfinished Columbaria!

Columbarium near Assen

Example / Exercise. Compute the determinant of a $1,000\times 1,000$ matrix in which 50 entries are not yet given.

Homework / Research Projects. • What with ZPUCs? • Use this to get an Alexander tangle invariant.

Reminders. {knots} ⇉ {matrices / quadratic forms} $\xrightarrow[\sigma]{\text{signature}}$ $\mathbb{Z}$:



$\to \sigma = \sigma_+ - \sigma_- + s$

Kashaev's Conj. [Ka] $\sigma_{Kas}=2\sigma_{TL}$.

	Tristram-Levine (*TL*)	Kashaev (*Kas*)
	$A \mathrel{+}= \begin{pmatrix} -r & -t & 2t & \bar t \\ -\bar t & 0 & \bar t & 0 \\ 2\bar t & t & -r & -\bar t \\ t & 0 & -t & 0 \end{pmatrix}\begin{matrix} i\\ j\\ k\\ l\end{matrix}$, $s=0$	$A \mathrel{+}= \begin{pmatrix} v & u & 1 & u \\ u & 1 & u & 1 \\ 1 & u & v & u \\ u & 1 & u & 1 \end{pmatrix}\begin{matrix} i\\ j\\ k\\ l\end{matrix}$, $s \mathrel{-}= 1$
	$A \mathrel{+}= \begin{pmatrix} r & -t & -2\bar t & \bar t \\ -\bar t & 0 & \bar t & 0 \\ -2t & t & r & -\bar t \\ t & 0 & -t & 0 \end{pmatrix}\begin{matrix} i\\ j\\ k\\ l\end{matrix}$, $s=0$	$A \mathrel{-}= \begin{pmatrix} v & u & 1 & u \\ u & 1 & u & 1 \\ 1 & u & v & u \\ u & 1 & u & 1 \end{pmatrix}\begin{matrix} i\\ j\\ k\\ l\end{matrix}$, $s \mathrel{+}= 1$

where $|\omega|=1$, $t=1-\omega$, $r=t+\bar t$, $v=\mathrm{Re}(\omega)$, and $u=\mathrm{Re}(\omega^{1/2})$.

A quadratic form on a v.s. V over $\mathbb{C}$ is a quadratic $Q\colon V\to\mathbb{C}$, or a sesquilinear Hermitian $\langle\cdot,\cdot\rangle$ on $V\times V$ (so $\langle x,y\rangle=\overline{\langle y,x\rangle}$ and $Q(y)=\langle y,y\rangle$), or given a basis η_i of V^*, a matrix $A=(a_{ij})$ with $A=\bar{A}^T$ and $Q=\sum a_{ij}\bar\eta_i\eta_j$. The signature σ of Q is $\sigma_+-\sigma_-$, where for some P, $\bar P^T A P = \mathrm{diag}(1,\overset{\sigma_+}{\cdots},1,-1,\overset{\sigma_-}{\cdots},-1,0,\ldots)$.

A *Partial Quadratic (PQ)* on V is a quadratic Q defined only on a subspace $\mathcal{D}_Q\subset V$. We add PQs with $\mathcal{D}_{Q_1+Q_2}:=\mathcal{D}_{Q_1}\cap\mathcal{D}_{Q_2}$. Given a linear $\psi\colon V\to W$ and a PQ Q on W, there is an obvious *pullback* ψ^*Q, a PQ on V.

Theorem 1. Given a linear $\phi\colon V\to W$ and a PQ Q on V, there is a unique *pushforward* PQ ϕ_*Q on W such that for every *PQ* U on W, $\sigma_V(Q+\phi^*U)=\sigma_{\ker\phi}(Q|_{\ker\phi})+\sigma_W(U+\phi_*Q)$.

(If you must, $\mathcal{D}(\phi_*Q)=\phi(\mathrm{ann}_Q(\mathcal{D}(Q)\cap\ker\phi))$ and $(\phi_*Q)(w)=Q(v)$, where v is s.t. $\phi(v)=w$ and $Q(v,\mathrm{rad}\,Q|_{\ker\phi})=0$).

Gist of the Proof.



... and the quadratic $F =: \phi_*Q$ is well-defined only on $D:=\ker C$.

Exactly what we want, if the Zombian is the signature!

V: The full space of *faces*.

W: The boundary, made of *gaps*.

Q: The known parts.

U: The part yet unknown.

$\sigma_V(Q+\phi^*(U))$: The overall Zombian.

$\sigma(Q|_{\ker\phi})$: An internal bit. $U+\phi_*Q$: A boundary bit.

And so our ZPUC is the pair $S=(\sigma(Q|_{\ker\phi}),\phi_*Q)$.



A *Shifted Partial Quadratic (SPQ)* on V is a pair $S=(s\in\mathbb{Z}, Q$ a PQ on $V)$. addition also adds the shifts, pullbacks keep the shifts, yet $\phi_*S:=(s+\sigma_{\ker\phi}(Q|_{\ker\phi}),\phi_*Q)$ and $\sigma(S):=s+\sigma(Q)$.

Theorem 1' (*Reciprocity*). Given $\phi\colon V\to W$, for SPQs S on V and U on W we have $\sigma_V(S+\phi^*U)=\sigma_W(U+\phi_*S)$ (and this characterizes ϕ_*S).

Theorem 2. ψ^* and ϕ_* are functorial. Also, if $\alpha /\!\!/ \beta = \gamma /\!\!/ \delta$, α is surjective, β is injective, and $\mathrm{im}\,\gamma\supset\ker\delta$, then $\gamma^* /\!\!/ \alpha_* = \delta_* /\!\!/ \beta^*$. Finally, ψ^* is additive but ϕ_* isn't.



Definition. $\mathcal{S}\left(\text{cyclic set } g_1, g_2, g_3, \ldots\right) := \left\{\begin{matrix}\text{SPQ } S\\ \text{on } \langle g_i\rangle\end{matrix}\right\}$.



Theorem 3. $\{\mathcal{S}(\text{cyclic sets})\}$ is a planar algebra, with compositions $\mathcal{S}(D)((S_i)):=\phi^D_*(\psi^*_D(\bigoplus_i S_i))$, where $\psi_D\colon\langle f_i\rangle\to\langle g_{\alpha i}\rangle$ maps every face of D to the sum of the input gaps adjacent to it and $\phi^D\colon\langle f_i\rangle\to\langle g_i\rangle$ maps every face to the sum of the output gaps adjacent to it. So for our D, ψ_D is $f_1\mapsto g_{34}$, $f_2\mapsto g_{31}+g_{14}+g_{24}+g_{33}$, $f_3\mapsto g_{32}$, $f_4\mapsto g_{11}$, $f_5\mapsto g_{13}+g_{21}$, $f_6\mapsto g_{23}$, $f_7\mapsto g_{12}+g_{22}$ and ϕ^D is $f_1\mapsto g_1$, $f_2\mapsto g_2+g_6$, $f_3\mapsto 0$, $f_4\mapsto g_3$, $f_5\mapsto 0$, $f_6\mapsto g_5$, $f_7\mapsto g_4$.



Connection Diagram

Theorem 4. TL and Kas, defined on X and $\bar X$ as before, extend to planar algebra morphisms {tangles} → $\{\mathcal{S}\}$.

Implementation (sources: http://drorbn.net/icerm23/ap). I like it most when the implementation matches the math perfectly. We failed here.

```
Once[<< KnotTheory`];
```

```
Loading KnotTheory` version
  of February 2, 2020, 10:53:45.2097.
Read more at http://katlas.org/wiki/KnotTheory.
```

Utilities. The step function, algebraic numbers, canonical forms.

```
θ[x_] /; NumericQ[x] := UnitStep[x]
```

```
ω2[v_][p_] := Module[{q = Expand[p], n, c},
   If[q === 0, 0,
    c = Coefficient[q, ω, n = Exponent[q, ω]];
    c v^n + ω2[v][q - c (ω + ω^-1)^n]]];
```

```
sign[ℰ_] := Module[{n, d, v, p, rs, e, k},
  {n, d} = NumeratorDenominator[ℰ];
  {n, d} /= ω^(Exponent[n,ω]/2+Exponent[n,ω,Min]/2);
  p = Factor[ω2[v]@n * ω2[v]@d /. v → 4 u^2 - 2];
  rs = Solve[p == 0, u, Reals];
  If[rs === {}, Sign[p /. u → 0],
   rs = Union@(u /. rs);
   Sign[(-1)^(e=Exponent[p,u]) Coefficient[p, u, e]] + Sum[
     k = 0;
     While[(d = RootReduce[∂_{u,++k} p /. u → r]) == 0];
     If[EvenQ[k], 0, 2 Sign[d]] * θ[u - r],
     {r, rs}]
  ]
 ]
```

```
SetAttributes[B, Orderless];
CF[b_B] := RotateLeft[#, First@Ordering[#] - 1] & /@
  DeleteCases[b, {}]
```

```
CF[ℰ_] := Module[{γs = Union@Cases[ℰ, γ_ | γ̄_, ∞]},
  Total[CoefficientRules[ℰ, γs] /.
    (ps_ → c_) :→ Factor[c] × Times @@ γs^ps] ]
```

```
CF[{}] = {};
CF[𝒞_List] :=
 Module[{γs = Union@Cases[𝒞, γ_, ∞], γ},
  CF /@ DeleteCases[0][
    RowReduce[Table[∂_γ r, {r, 𝒞}, {γ, γs}]].γs] ]
```

```
(ℰ_)* := ℰ /. {γ̄ → γ, γ → γ̄, ω → ω^-1, c_Complex :→ c*};
r_Rule⁺ := {r, r*}
```

```
RulesOf[γi_ + rest_.] := (γi → -rest)⁺;
CF[PQ[𝒞_, q_]] := Module[{n𝒞 = CF[𝒞]},
  PQ[n𝒞, CF[q /. Union @@ RulesOf /@ n𝒞]] ]
```

```
CF[Σb_[σ_, pq_]] := Σ_CF[b][σ, CF[pq]]
```

Pretty-Printing.

```
Format[Σb_B[σ_, PQ[𝒞_, q_]]] := Module[{γs},
  γs = γ# & /@ Join @@ b;
  Column[{TraditionalForm@σ,
    TableForm[Join[
      Prepend[""] /@ Table[TraditionalForm[∂_c r],
        {r, 𝒞}, {c, γs}],
      {Prepend[""][
        Join @@
         (b /. {l_, m___, r_} :→
            {DisplayForm@RowBox[{"(", l}],
             m, DisplayForm@RowBox[{r, ")"}]}) /.
        i_Integer :→ γi ]},
      MapThread[Prepend,
       {Table[TraditionalForm[∂_{r,c} q], {r, γs*},
         {c, γs}], γs*}]
     ], TableAlignments → Center]
   }, Center]];
```

The Face-Centric Core.

```
Σb1_[σ1_, PQ[𝒞1_, q1_]] ⊕ Σb2_[σ2_, PQ[𝒞2_, q2_]] ^:=
  CF@Σ_Join[b1,b2][σ1 + σ2, PQ[𝒞1 ⋃ 𝒞2, q1 + q2]];
```

GT for Gap Touch: 


```
GT_{i_,j_}@Σ_B[{li___,i_,ri___},{lj___,j_,rj___},bs___][σ_,
   PQ[𝒞_, q_]] :=
  CF@Σ_B[{ri,li,j,rj,lj,i},bs][σ, PQ[𝒞 ⋃ {γi - γj}, q]]
```



cor·don (kôr′dn)

n.

1. A line of people, military posts, or ships stationed around an area to enclose or guard it: *a police cordon*.

2. A rope, line, tape, or similar border stretched around an area, usually by the police, indicating that access is restricted.

$$\begin{matrix} s \\ i \end{matrix}\begin{pmatrix} 0 & \phi\, C_{\text{rest}} \\ \bar\phi^T & \lambda \quad \theta \\ \bar C^T_{\text{rest}} & \bar\theta^T\, A_{\text{rest}} \end{pmatrix} \to \begin{cases} \exists p\ \phi_p \neq 0 & \text{use } \phi_p \text{ to kill its row and column, drop a } \begin{pmatrix} 01 \\ 10 \end{pmatrix} \text{ summand} \\ \phi=0, \lambda\neq 0 & \text{use } \lambda \text{ to kill } \theta\text{, let } s \mathrel{+}= \operatorname{sign}(\lambda) \\ \phi=0, \lambda=0 & \text{append } \theta \text{ to } C_{\text{rest}}. \end{cases}$$

```
Cordon_{i_}@Σ_B[{li___,i_,ri___},bs___][σ_, PQ[𝒞_, q_]] :=
 Module[{ϕ = ∂_γi 𝒞, λ = ∂_{γ̄i,γi} q, nσ = σ, n𝒞, nq, p},
  {p} = FirstPosition[(# =!= 0) & /@ ϕ, True, {0}];
  {n𝒞, nq} = Which[
    p > 0, {𝒞, q} /. (γi → -𝒞⟦p⟧ / ϕ⟦p⟧)⁺ /. (γi → 0)⁺,
    λ =!= 0, (nσ += sign[λ];
      {𝒞, q /. (γi → -(∂_γ̄i q) / λ)⁺ /. (γi → 0)⁺}),
    λ === 0, {𝒞 ⋃ {∂_γ̄i q}, q /. (γi → 0)⁺}];
  CF@Σ_B[Most@{ri,li},bs][nσ,
    PQ[n𝒞, nq] /. (γ_Last@{ri,li} → γ_First@{ri,li})⁺] ]
```

Strand Operations. c for contract, mc for magnetic contract:

```
c_{i_,j_}@t : Σ_B[{li___,i_,ri___},{___,j_,___},___][__] :=
  t // GT_{j,First@{ri,li}} // Cordon_j
```

```
c_{i_,j_}@t : Σ_B[{___,i_,j_,___},___][__] := Cordon_j@t
c_{i_,j_}@t : Σ_B[{j_,___,i_},___][__] := Cordon_j@t
c_{i_,j_}@t : Σ_B[{___,j_,i_,___},___][__] := Cordon_i@t
c_{i_,j_}@t : Σ_B[{i_,___,j_},___][__] := Cordon_i@t
```

```
mc[ℰ_] := ℰ //.
  t : Σ_B[{___,i_,___},{___,j_,___},___][__] |
     Σ_B[{___,i_,j_,___},___][__] | Σ_B[{j_,___,i_},___][__] /;
    i + j == 0 :> c_{i,j}@t
```

The Crossings (and empty strands).

```
Kas@P_{i_,j_} := CF@Σ_B[{i,j}][0, PQ[{}, 0]];
TL@P_{i_,j_} := CF@Σ_B[{i,j}][0, PQ[{}, 0]]
```

```
Kas[x : X[i_, j_, k_, l_]] :=
  Kas@If[PositiveQ[x], X_{-i,j,k,-l}, X̄_{-j,k,l,-i}];
Kas[(x : X | X̄)_{fs__}] := Module[{v = 2 u^2 - 1, p, γs, m},
  γs = γ_# & /@ {fs}; p = (x === X);
  m = If[p, ( v u 1 u ), - ( v u 1 u )];
            ( u 1 u 1 )     ( u 1 u 1 )
            ( 1 u v u )     ( 1 u v u )
            ( u 1 u 1 )     ( u 1 u 1 )
  CF@Σ_B[{fs}][If[p, -1, 1], PQ[{}, γs*.m.γs]]]
```

```
TL[x : X[i_, j_, k_, l_]] :=
  TL@If[PositiveQ[x], X_{-i,j,k,-l}, X̄_{-j,k,l,-i}];
TL[(x : X | X̄)_{fs__}] := Module[{t = 1 - ω, r, γs, m},
  r = t + t*; γs = γ_# & /@ {fs};
  m = If[x === X,
     ( -r   -t   2t   t*  )   ( r    -t   -2t*  t*  )
     ( -t*   0   t*   0   ) , ( -t*   0    t*   0   ) ];
     ( 2t*   t   -r   -t* )   ( -2t   t    r    -t* )
     ( t     0   -t   0   )   ( t     0    -t   0   )
  CF@Σ_B[{fs}][0, PQ[{}, γs*.m.γs]]]
```

Evaluation on Tangles and Knots.

```
Kas[K_] := Fold[mc[#1 ⊕ #2] &, Σ_B[][0, PQ[{}, 0]],
    List @@ (Kas /@ PD@K)];
KasSig[K_] := Expand[Kas[K]⟦1⟧ / 2]
```

```
TL[K_] :=
  Fold[mc[#1 ⊕ #2] &, Σ_B[][0, PQ[{}, 0]],
    List @@ (TL /@ PD@K)] /.
   θ[c_ + u] /; Abs[c] ≥ 1 :> θ[c];
TLSig[K_] := TL[K]⟦1⟧
```

Reidemeister 3.




```
R3L = PD[X_{-2,5,4,-1}, X_{-3,7,6,-5},
    X_{-6,9,8,-4}];
R3R = PD[X_{-3,5,4,-2}, X_{-4,6,8,-1},
    X_{-5,7,9,-6}];
{TL@R3L == TL@R3R, Kas@R3L == Kas@R3R}
```

{True, True}

```
Kas@R3L
```

$$2\,\theta\left(u-\tfrac{1}{2}\right)-2\,\theta\left(u+\tfrac{1}{2}\right)-2$$

	$(\gamma_{-3}$	γ_7	γ_9	γ_8	γ_{-1}	$\gamma_{-2})$
$\bar\gamma_{-3}$	$\frac{2u^2(4u^2-3)}{(2u-1)(2u+1)}$	$\frac{u(4u^2-3)}{(2u-1)(2u+1)}$	$-\frac{1}{(2u-1)(2u+1)}$	$-\frac{2u}{(2u-1)(2u+1)}$	$-\frac{1}{(2u-1)(2u+1)}$	$\frac{u(4u^2-3)}{(2u-1)(2u+1)}$
$\bar\gamma_{7}$	$\frac{u(4u^2-3)}{(2u-1)(2u+1)}$	$\frac{2(2u^2-1)}{(2u-1)(2u+1)}$	$\frac{u(4u^2-3)}{(2u-1)(2u+1)}$	$-\frac{1}{(2u-1)(2u+1)}$	$-\frac{2u}{(2u-1)(2u+1)}$	$-\frac{1}{(2u-1)(2u+1)}$
$\bar\gamma_{9}$	$-\frac{1}{(2u-1)(2u+1)}$	$\frac{u(4u^2-3)}{(2u-1)(2u+1)}$	$\frac{2u^2(4u^2-3)}{(2u-1)(2u+1)}$	$\frac{u(4u^2-3)}{(2u-1)(2u+1)}$	$-\frac{1}{(2u-1)(2u+1)}$	$-\frac{2u}{(2u-1)(2u+1)}$
$\bar\gamma_{8}$	$-\frac{2u}{(2u-1)(2u+1)}$	$-\frac{1}{(2u-1)(2u+1)}$	$\frac{u(4u^2-3)}{(2u-1)(2u+1)}$	$\frac{2u^2(4u^2-3)}{(2u-1)(2u+1)}$	$\frac{u(4u^2-3)}{(2u-1)(2u+1)}$	$-\frac{1}{(2u-1)(2u+1)}$
$\bar\gamma_{-1}$	$-\frac{1}{(2u-1)(2u+1)}$	$-\frac{2u}{(2u-1)(2u+1)}$	$-\frac{1}{(2u-1)(2u+1)}$	$\frac{u(4u^2-3)}{(2u-1)(2u+1)}$	$\frac{2(2u^2-1)}{(2u-1)(2u+1)}$	$\frac{u(4u^2-3)}{(2u-1)(2u+1)}$
$\bar\gamma_{-2}$	$\frac{u(4u^2-3)}{(2u-1)(2u+1)}$	$-\frac{1}{(2u-1)(2u+1)}$	$-\frac{2u}{(2u-1)(2u+1)}$	$-\frac{1}{(2u-1)(2u+1)}$	$\frac{u(4u^2-3)}{(2u-1)(2u+1)}$	$\frac{2u^2(4u^2-3)}{(2u-1)(2u+1)}$

Reidemeister 2.




```
TL@PD[X_{-2,4,3,-1}, X̄_{-4,6,5,-3}]
```

0

	1	0	−1	0
	$(\gamma_{-2}$	γ_6	γ_5	$\gamma_{-1})$
$\bar\gamma_{-2}$	0	0	0	0
$\bar\gamma_{6}$	0	0	0	0
$\bar\gamma_{5}$	0	0	0	0
$\bar\gamma_{-1}$	0	0	0	0

```
{TL@PD[X_{-2,4,3,-1}, X̄_{-4,6,5,-3}] == GT_{5,-2}@TL@PD[P_{-1,5}, P_{-2,6}],
 Kas@PD[X_{-2,4,3,-1}, X̄_{-4,6,5,-3}] == GT_{5,-2}@Kas@PD[P_{-1,5}, P_{-2,6}]}
```

{True, True}

Reidemeister 1.




```
{TL@PD[X_{-3,3,2,-1}] == TL@P_{-1,2},
 Kas@PD[X_{-3,3,2,-1}] == Kas@P_{-1,2}}
```

{True, True}

A Knot.



```
f = TLSig[Knot[8, 5]]
```

$$2\,\theta\left[-\frac{\sqrt{3}}{2}+u\right]-2\,\theta\left[\frac{\sqrt{3}}{2}+u\right]-2\,\theta\left[u-(-0.630\ldots)\right]+2\,\theta\left[u-(0.630\ldots)\right]$$

```
Plot[f, {u, -1, 1}]
```



The Conway-Kinoshita-Terasaka Tangles.







```
T1 = PD[X̄-6,2,7,-1, X̄-2,8,3,-7,
    X̄-8,4,9,-3, X-11,6,12,-5,
    X-4,11,5,-10];
T2 = PD[X-6,2,7,-1, X-2,8,3,-7,
    X-8,4,9,-3, X̄-12,6,13,-5,
    X̄-4,12,5,-11, X̄-10,15,11,-14, X̄-15,10,16,-9];
```

```
Column@{TL[T1], Kas[T1]}
```

$$-2\,\Theta\left(u-\frac{\sqrt{3}}{2}\right)+2\,\Theta\left(u+\frac{\sqrt{3}}{2}\right)-1$$

	γ_{-10}	γ_{9}	γ_{-1}	γ_{12}
$\bar\gamma_{-10}$	0	$1-\omega$	0	$\omega-1$
$\bar\gamma_{9}$	$\frac{\omega-1}{\omega}$	$\frac{2\omega}{\omega^2-\omega+1}$	$-\frac{\omega-1}{\omega}$	$-\frac{2\omega}{\omega^2-\omega+1}$
$\bar\gamma_{-1}$	0	$\omega-1$	0	$1-\omega$
$\bar\gamma_{12}$	$-\frac{\omega-1}{\omega}$	$-\frac{2\omega}{\omega^2-\omega+1}$	$\frac{\omega-1}{\omega}$	$\frac{2\omega}{\omega^2-\omega+1}$

$$-2\,\Theta\left(u-\frac{\sqrt{3}}{2}\right)+2\,\Theta\left(u+\frac{\sqrt{3}}{2}\right)-1$$

	γ_{-10}	γ_{9}	γ_{-1}	γ_{12}
$\bar\gamma_{-10}$	$2(u-1)(u+1)(4u^2-3)$	0	$-2(u-1)(u+1)(4u^2-3)$	0
$\bar\gamma_{9}$	0	$\frac{1}{2(4u^2-3)}$	0	$-\frac{1}{2(4u^2-3)}$
$\bar\gamma_{-1}$	$-2(u-1)(u+1)(4u^2-3)$	0	$2(u-1)(u+1)(4u^2-3)$	0
$\bar\gamma_{12}$	0	$-\frac{1}{2(4u^2-3)}$	0	$\frac{1}{2(4u^2-3)}$

```
Column@{TL[T2], Kas[T2]}
```

$$0$$

	γ_{-14}	γ_{16}	γ_{-1}	γ_{13}
$\bar\gamma_{-14}$	0	$1-\omega$	0	$\omega-1$
$\bar\gamma_{16}$	$\frac{\omega-1}{\omega}$	$-\frac{2(\omega-1)^2\omega}{\omega^4-3\omega^3+5\omega^2-3\omega+1}$	$-\frac{\omega-1}{\omega}$	$\frac{2(\omega-1)^2\omega}{\omega^4-3\omega^3+5\omega^2-3\omega+1}$
$\bar\gamma_{-1}$	0	$\omega-1$	0	$1-\omega$
$\bar\gamma_{13}$	$-\frac{\omega-1}{\omega}$	$\frac{2(\omega-1)^2\omega}{\omega^4-3\omega^3+5\omega^2-3\omega+1}$	$\frac{\omega-1}{\omega}$	$-\frac{2(\omega-1)^2\omega}{\omega^4-3\omega^3+5\omega^2-3\omega+1}$

$$1$$

	γ_{-14}	γ_{16}	γ_{-1}	γ_{13}
$\bar\gamma_{-14}$	$\frac{1}{2}(-16u^4+28u^2-13)$	0	$\frac{1}{2}(16u^4-28u^2+13)$	0
$\bar\gamma_{16}$	0	$-\frac{2(u-1)(u+1)}{16u^4-28u^2+13}$	0	$\frac{2(u-1)(u+1)}{16u^4-28u^2+13}$
$\bar\gamma_{-1}$	$\frac{1}{2}(16u^4-28u^2+13)$	0	$\frac{1}{2}(-16u^4+28u^2-13)$	0
$\bar\gamma_{13}$	0	$\frac{2(u-1)(u+1)}{16u^4-28u^2+13}$	0	$-\frac{2(u-1)(u+1)}{16u^4-28u^2+13}$

Examples with non-trivial co-dimension.





```
B1 = PD[X-5,2,6,-1, X̄-8,3,9,-2,
    X-11,4,12,-3, X-12,10,13,-9,
    X̄-13,7,14,-6];
B2 = PD[X-5,2,6,-1, X̄-9,3,10,-2,
    X-10,7,11,-6, X̄-12,4,13,-3, X-13,8,14,-7];
```

```
Column@{TL[B1], Kas[B1]}
```

$$0$$

	γ_{-11}	γ_{4}	γ_{10}	γ_{7}	γ_{14}	γ_{-1}	γ_{-5}	γ_{-8}
	1	0	-1	0	$\frac{1}{\omega}$	0	$-\frac{1}{\omega}$	0
	0	0	0	-1	$\frac{1}{\omega}$	0	$-\frac{1}{\omega}$	1
$\bar\gamma_{-11}$	0	0	0	0	0	0	0	0
$\bar\gamma_{4}$	0	0	0	0	$\frac{\omega-1}{\omega^2}$	0	$-\frac{\omega-1}{\omega^2}$	0
$\bar\gamma_{10}$	0	0	0	0	$-\frac{\omega-1}{\omega}$	0	$\frac{\omega-1}{\omega}$	0
$\bar\gamma_{7}$	0	0	0	0	$\frac{(\omega-1)^2}{\omega^2}$	0	$-\frac{(\omega-1)^2}{\omega^2}$	0
$\bar\gamma_{14}$	0	$-((\omega-1)\,\omega)$	$\omega-1$	$(\omega-1)^2$	0	$-\frac{\omega-1}{\omega}$	$\frac{\omega-1}{\omega}$	0
$\bar\gamma_{-1}$	0	0	0	0	$\omega-1$	0	$1-\omega$	0
$\bar\gamma_{-5}$	0	$(\omega-1)\,\omega$	$1-\omega$	$-(\omega-1)^2$	$1-\omega$	$\frac{\omega-1}{\omega}$	$\frac{(\omega-1)^2}{\omega}$	0
$\bar\gamma_{-8}$	0	0	0	0	0	0	0	0

$$0$$

	γ_{-11}	γ_{4}	γ_{10}	γ_{7}	γ_{14}	γ_{-1}	γ_{-5}	γ_{-8}
	1	0	-1	0	1	0	-1	0
$\bar\gamma_{-11}$	0	0	0	0	0	0	0	0
$\bar\gamma_{4}$	0	0	0	-1	$-u$	0	u	1
$\bar\gamma_{10}$	0	0	0	$-u$	$1-2u^2$	0	$2u^2-1$	u
$\bar\gamma_{7}$	0	-1	$-u$	$2u^2-3$	$-u$	-1	0	1
$\bar\gamma_{14}$	0	$-u$	$1-2u^2$	$-u$	-1	$-u$	$-2(u-1)(u+1)$	u
$\bar\gamma_{-1}$	0	0	0	-1	$-u$	0	u	1
$\bar\gamma_{-5}$	0	u	$2u^2-1$	0	$-2(u-1)(u+1)$	u	$4u^2-3$	0
$\bar\gamma_{-8}$	0	1	u	1	u	1	0	$1-2u^2$

```
Column@{TL[B2], Kas[B2]}
```

$$0$$

	γ_{-12}	γ_{4}	γ_{8}	γ_{14}	γ_{11}	γ_{-1}	γ_{-5}	γ_{-9}
$\bar\gamma_{-12}$	$\frac{(\omega-1)^2}{\omega}$	$\omega-1$	$-2(\omega-1)$	$\frac{2(\omega-1)^2}{\omega}$	$\frac{2(\omega-1)}{\omega^2}$	0	$-\frac{2(\omega-1)}{\omega^2}$	$-\frac{(\omega-1)(2\omega-3)}{\omega}$
$\bar\gamma_{4}$	$-\frac{\omega-1}{\omega}$	0	$\frac{\omega-1}{\omega}$	0	0	0	0	0
$\bar\gamma_{8}$	$\frac{2(\omega-1)}{\omega}$	$1-\omega$	$\frac{(\omega-1)^2}{\omega}$	$-\frac{(\omega-1)(2\omega-3)}{\omega}$	$-\frac{2(\omega-1)}{\omega^2}$	0	$\frac{2(\omega-1)}{\omega^2}$	$\frac{2(\omega-2)(\omega-1)}{\omega}$
$\bar\gamma_{14}$	$\frac{2(\omega-1)^2}{\omega}$	0	$-\frac{(\omega-1)(3\omega-2)}{\omega}$	$\frac{3(\omega-1)^2}{\omega}$	$-\frac{(\omega-2)(\omega-1)}{\omega^2}$	0	$-\frac{2(\omega-1)}{\omega^2}$	$-\frac{2(\omega-2)(\omega-1)}{\omega}$
$\bar\gamma_{11}$	$-2(\omega-1)\,\omega$	0	$2(\omega-1)\,\omega$	$-((\omega-1)(2\omega-1))$	$\frac{(\omega-1)^2}{\omega}$	$-\frac{\omega-1}{\omega}$	$\frac{2(\omega-1)}{\omega}$	$2(\omega-1)^2$
$\bar\gamma_{-1}$	0	0	0	0	$\omega-1$	0	$1-\omega$	0
$\bar\gamma_{-5}$	$2(\omega-1)\,\omega$	0	$-2(\omega-1)\,\omega$	$2(\omega-1)\,\omega$	$-2(\omega-1)$	$\frac{\omega-1}{\omega}$	$\frac{(\omega-1)^2}{\omega}$	$-((\omega-1)(2\omega-1))$
$\bar\gamma_{-9}$	$-\frac{(\omega-1)(3\omega-2)}{\omega}$	0	$\frac{2(\omega-1)(2\omega-1)}{\omega}$	$-\frac{2(\omega-1)(2\omega-1)}{\omega}$	$\frac{2(\omega-1)^2}{\omega^2}$	0	$-\frac{(\omega-2)(\omega-1)}{\omega^2}$	$\frac{3(\omega-1)^2}{\omega}$

$$2\,\Theta\left(u-\frac{\sqrt{3}}{2}\right)-2\,\Theta\left(u+\frac{\sqrt{3}}{2}\right)$$

	γ_{-12}	γ_{4}	γ_{8}	γ_{14}	γ_{11}	γ_{-1}	γ_{-5}	γ_{-9}
	1	$\frac{1}{2u}$	0	$-\frac{1}{2u}$	-1	$-\frac{1}{2u}$	0	$\frac{1}{2u}$
$\bar\gamma_{-12}$	0	0	0	0	0	0	0	0
$\bar\gamma_{4}$	0	$-\frac{(2u-1)(2u+1)(2u^2-1)}{4u^2(4u^2-3)}$	$-\frac{2u^2-1}{2u}$	$\frac{1}{4u^2(4u^2-3)}$	0	$-\frac{(2u-1)(2u+1)}{4u^2(4u^2-3)}$	$-\frac{1}{2u(4u^2-3)}$	$\frac{8u^4-6u^2-1}{4u^2(4u^2-3)}$
$\bar\gamma_{8}$	0	$-\frac{2u^2-1}{2u}$	$-2(u-1)(u+1)$	$\frac{2u^2-1}{2u}$	0	$-\frac{1}{2u}$	0	$\frac{1}{2u}$
$\bar\gamma_{14}$	0	$\frac{1}{4u^2(4u^2-3)}$	$\frac{2u^2-1}{2u}$	$\frac{(2u^2-1)(16u^4-16u^2+1)}{4u^2(4u^2-3)}$	0	$-\frac{8u^4-10u^2+1}{4u^2(4u^2-3)}$	$\frac{1}{2u(4u^2-3)}$	$\frac{1}{4u^2(4u^2-3)}$
$\bar\gamma_{11}$	0	0	0	0	0	0	0	0
$\bar\gamma_{-1}$	0	$-\frac{(2u-1)(2u+1)}{4u^2(4u^2-3)}$	$-\frac{1}{2u}$	$-\frac{8u^4-10u^2+1}{4u^2(4u^2-3)}$	0	$\frac{8u^4-10u^2-1}{4u^2(4u^2-3)}$	$\frac{8u^4-10u^2+1}{2u(4u^2-3)}$	$\frac{16u^4-16u^2+1}{4u^2(4u^2-3)}$
$\bar\gamma_{-5}$	0	$-\frac{1}{2u(4u^2-3)}$	0	$\frac{1}{2u(4u^2-3)}$	0	$\frac{8u^4-10u^2+1}{2u(4u^2-3)}$	$\frac{2(u-1)(u+1)(2u-1)(2u+1)}{4u^2-3}$	$\frac{8u^4-6u^2-1}{2u(4u^2-3)}$
$\bar\gamma_{-9}$	0	$\frac{8u^4-6u^2-1}{4u^2(4u^2-3)}$	$\frac{1}{2u}$	$\frac{1}{4u^2(4u^2-3)}$	0	$\frac{16u^4-16u^2+1}{4u^2(4u^2-3)}$	$\frac{8u^4-6u^2-1}{2u(4u^2-3)}$	$-\frac{32u^6-64u^4+30u^2+1}{4u^2(4u^2-3)}$

$$\begin{pmatrix}A & B\\ C & U\end{pmatrix}\xrightarrow{\det(A)}\begin{pmatrix}I & A^{-1}B\\ C & U\end{pmatrix}\xrightarrow{1}\begin{pmatrix}I & A^{-1}B\\ 0 & U-CA^{-1}B\end{pmatrix},$$

so $\det\begin{pmatrix}A & B\\ C & U\end{pmatrix}=\det(A)\det(U-CA^{-1}B)$. (what if $\nexists A^{-1}$?)

Questions. 1. Does this have a topological meaning? 2. Prove the Kashaev conjecture. Is there a version for tangles? 3. Find all solutions of R123 in our "algebra". 4. Braids and the Burau representation. 5. Recover the work in "Prior Art". 6. Are there any concordance properties? 7. What is the "SPQ group"? 8. The jumping points of signatures are the roots of the Alexander polynomial. Does this generalize to tangles? 9. Which of the three Cordon cases is the most common? 10. Are there interesting examples of tangles for which rels is non-trivial? 11. Is the *pq* part determined by Γ-calculus? 12. Is the *pq* part determined by finite type invariants? 13. Does it work with closed components / links? 14. Strand-doubling formulas? 15. A multivariable version? 16. Mutation invariance? 17. Ribbon knots? 18. Are there "face-virtual knots"? 19. Does the pushforward story extend to ranks? To formal Gaussian measures? To super Gaussian measures?

References.

[CC] D. Cimasoni, A. Conway, *Colored Tangles and Signatures,* Math. Proc. Camb. Phil. Soc. **164** (2018) 493–530, arXiv:1507.07818.

[Co] A. Conway, *The Levine-Tristram Signature: A Survey,* arXiv:1903.04477.

[GG] J-M. Gambaudo, É. Ghys, *Braids and Signatures,* Bull. Soc. Math. France **133-4** (2005) 541–579.

[Ka] R. Kashaev, *On Symmetric Matrices Associated with Oriented Link Diagrams,* in *Topology and Geometry, A Collection of Essays Dedicated to Vladimir G. Turaev,* EMS Press 2021, arXiv:1801.04632.

[Me] A. Merz, *An Extension of a Theorem by Cimasoni and Conway,* arXiv:2104.02993.

Acknowledgement. This work was partially supported by NSERC grant RGPIN-2018-04350 and by the Chu Family Foundation (NYC).

Proof of Theorem 1.

Uniqueness: If A and B are 2 pushforwards, then $\sigma_W(U+A) = \sigma_W(U+B)$ for all PQs U on W.

Thus $\mathcal{D}_A = \mathcal{D}_B$, because otherwise if $w \in \mathcal{D}_A \setminus \mathcal{D}_B$, by taking $U(w) = 1$ on $\mathcal{D}_U = \text{span}\{w\}$, we get $\sigma_W(U+A) = 1 \neq 0 = \sigma_W(U+B)$. Furthermore, A and B must agree where they are both defined, because by taking $U(w) = \frac{-A(w)-B(w)}{2}$ on $\mathcal{D}_U = \text{span}\{w\}$ we get $(U+A)(w) = \frac{A(w)-B(w)}{2} = -(U+B)(w)$, so we must have $A(w) = B(w)$ to satisfy $\sigma_W(U+A) = \sigma_W(U+B)$.

Existence: Define $\phi_* Q$ by $\mathcal{D}_{\phi_* Q} = \phi(\text{ann}_Q(\ker\phi))$ and $\phi_* Q(w) = Q(v)$ where $v \in \text{ann}_Q(\ker\phi)$. Note that $\phi_* Q$ is well-defined.

First consider when $U = 0$ on all of W. Let K be a maximal non-degenerate subspace of $\ker\phi$. Then $Q = Q|_K \oplus Q|_{\text{ann}_Q(K)}$, and we can write $\text{ann}_Q(K) = R \oplus A \oplus B$ where $R = \text{rad}_Q(\ker\phi)$ and A, B are chosen so that $A \subseteq \text{ann}_Q(R)$ and $B \subseteq \text{ann}_Q(K) \setminus \text{ann}_Q(R)$. Since $Q : R \to B^*$ is surjective, for any $v \in \mathcal{D}_Q$ there is some $r_v \in R$ such that $Q(r_v, B) = Q(v, B)$. If we choose the r_v so that $r_{v_1} + r_{v_2} = r_{v_1+v_2}$, then we can replace A by $A' = \{a - r_a : a \in A\}$ and B by $B' = \{b - \frac{1}{2} r_b : b \in B\}$ to get $Q = Q|_K \oplus Q|_{R\oplus B'} \oplus Q|_{A'}$.

Then notice that

- $\sigma_V(Q|_K) = \sigma_{\ker\phi}(Q|_{\ker\phi})$
- $\sigma_V(Q|_{R\oplus B'}) = 0$
- $\sigma_V(Q|_{A'}) = \sigma_W(\phi_* Q)$

so we get $\sigma_V(Q) = \sigma_{\ker\phi}(Q|_{\ker\phi}) + \sigma_W(\phi_* Q)$.

Now for an arbitrary U, note that $(Q + \phi^* U)|_{\ker\phi} = Q|_{\ker\phi}$ and $\phi_*(Q + \phi^* U) = \phi_* Q + U$ so we can replace Q in the $U = 0$ case by $Q + \phi^* U$ to get the general case.

Proof of Theorem 2.

It's clear that pullback is functorial and that pushforward by the identity is the identity. To show $(\phi\psi)_* = \phi_*\psi_*$, use theorem 1 repeatedly to get

$$
\begin{aligned}
&\sigma((\phi\psi)_* Q + U)\\
=&\sigma(Q + (\phi\psi)^* U)\\
=&\sigma(Q + \psi^*\phi^* U) - \sigma(Q|_{\ker\phi\psi})\\
=&\sigma(\psi_* Q + \phi^* U) + \sigma(Q|_{\ker\psi}) - \sigma(Q|_{\ker\phi\psi})\\
=&\sigma(\phi_*\psi_* Q + U) + \sigma(Q|_{\ker\psi}) + \sigma(\psi_* Q|_{\ker\phi}) - \sigma(Q|_{\ker\phi\psi})\\
=&\sigma(\phi_*\psi_* Q + U)
\end{aligned}
$$

for any U, where the last step uses theorem 1 on $Q|_{\phi\psi}$ with the map $\psi : \ker\phi\psi \to \ker\phi$.



To show $\alpha_*\gamma^* = \beta^*\delta_*$, first note that $\beta^*\beta_*$ is the identity on any PQ since β is injective, so

$$\alpha_*\gamma^* Q = \beta^*(\beta\alpha)_*\gamma^* Q = \beta^*(\delta\gamma)_*\gamma^* Q = \beta^*\delta_*\gamma_*\gamma^* Q$$

As $\beta^*\delta_*\gamma_*\gamma^* Q$ and $\beta^*\delta_* Q$ have the same values where they are both defined, it remains to show that they have the same domain. Since α is surjective and γ is surjective onto $\ker(\delta)$, we see that

$$\beta^{-1}\delta(A) = \beta^{-1}\delta(A \cap \text{im}\,\gamma)$$

for any subspace A. By taking $A = \text{ann}_Q(\ker\delta)$, the two sides of the equality become the domains of $\beta^*\delta_* Q$ and $\beta^*\delta_*\gamma_*\gamma^* Q$.